

Motorola Semiconductor Application Note

AN-HK-33

In-Circuit Programming of FLASH Memory in the MC68HC908JL3

By Roger Fan
Applications Engineering
Microcontroller Division
Hong Kong

This application note describes a simple method of In-Circuit Programming (ICP) the FLASH memory in the Motorola MC68HC908JL3 (JL3) microcontroller and its variants: MC68HRC908JL3, MC68HC908JK3, MC68HRC908JK3, MC68HC908JK1, and MC68HRC908JK1.

For detailed specification on MC68HC908JL3 and its variants, please refer to the datasheet, Motorola order number MC68HC908JL3/H.

Introduction

The 4k-bytes JL3 FLASH memory can be programmed or erased using routines running either in User mode or Monitor mode. A high voltage input is not required by the JL3 for the program or erase operation; as it is generated by an internal charge-pump.

In-circuit programming is a process by which the device is programmed or erased with the device on the final circuit board (the target system). This ability allows the user code to be changed without having to remove the device off the target system for initial programming or reprogramming.

The ICP method that will be described is low-cost; with minimal system and user code changes. It involves two steps:

1. Erasing the entire FLASH memory in User mode.
2. Programming the FLASH memory in Monitor mode using Motorola's SPGMR08 Serial Programmer.

User mode

In the user operating mode, the MCU is running the user code (user program), that has been programmed in the FLASH memory. This is the mode in which the MCU will be running during normal operation in the target system.

Monitor mode

In the monitor operating mode, the MCU is running code that has been permanently programmed into an area of memory in the MCU during production. The monitor code consists of routines for communicating to a host connected using a serial link. The monitor code is the program that the MCU runs when the FLASH memory is blank (unprogrammed).

Mass Erasing the FLASH Memory in User Mode

The JL3 has the ability to erase the contents of its FLASH memory by executing code in its RAM area. The prerequisite for this operation is that an *erase routine* needs to be part of the user code in FLASH memory, which will be loaded to the RAM area when called by the main user program. On the JL3, an erase subroutine is available in the monitor ROM area. This subroutine is called after writing the two control bytes to the RAM locations \$0088 and \$0089.

An erase sequence can then be initiated by the MCU by monitoring some port pins for specific logic levels.

In this implementation, PTB0 and PTD3 are used for setting up a request for mass erase operation. After a reset, the main user program polls these two pins; if PTB0 = 1 and PTD3 = 0 (see figure 1), the main program will load the mass erase routine into its RAM and perform a FLASH mass erase operation.

In the program listing at the back of this application note, the erase routine delay timing is based on a bus frequency of 2.4576MHz. The mass erase operation is repeated until the user vectors and the security bytes are erased. The time required for the mass erase operation is less than two seconds.




Figure 1. Mass Erase Port Pin Configuration

The flowchart in figure 2 shows the sequence of events for the mass erase operation.




Figure 2. Mass Erase Flowchart

Procedure for mass erase

Use this step-by-step procedure for erasing the JL3 FLASH memory in user mode:

1. Switch off the power to the target system.
2. Isolate port pins PTB0 and PTD3 from target system logic.
3. Set PTB0 to high via a pull-up resistor to V_{DD} .
4. Set PTD3 to ground directly to V_{SS} .
5. Switch on the power to the target system.
6. Wait 2 seconds.
7. Switch off power to the target system.
8. FLASH memory is now erased.

The next section describes the procedure for programming the JL3 FLASH memory. crystal frequency must be 9.8304MHz

Programming the FLASH Memory in Monitor Mode

Programming the JL3's blank FLASH memory is achieved by running the MCU in Monitor mode; and with a host connected using a serial link. Monitor mode can be entered in one of two ways after a power-on-reset:

- A high voltage ($1.5 \times V_{DD}$) applied to the $\overline{IRQ1}$ pin, or
- The FLASH memory is erased blank.

The latter method for entering monitor mode for programming the FLASH memory will be described here. With this method, the MCU enters monitor mode after a power-on reset when it detects that the reset vector, \$FFFE–\$FFFF, is blank (containing \$FF). Running the monitor mode in this way is called the *ICP mode*.

The Motorola *SPGMR08 serial programmer* is used as the interface between the target system and the PC host system.

Figure 2 shows the connection to the SPGMR08. Three lines are used:

- PTB0 — This is the serial data link between the host and the MCU in the target system.
- \overline{RST} — This line provides synchronization between the host and the MCU in the target system.
- GND — Common ground for the systems.

For this configuration, the bus frequency should be at 2.4576MHz for the 9600 baud communication between the MCU and host PC.




Figure 3. Programming Setup

Once the programming system is connected as in figure 3, and the systems are switched on (target system first), programming is carried out by running the *MCUscribe* utility supplied with the SPGMR08. When *MCUscribe* has finished programming, the "power-off" command on the screen-menu should be selected before power is switched off on the target system. The jumpers can then be set back to their original positions.

Further Information

The above ICP method has two limitations. They are:

1. The erase and program operations are for the entire 4k-bytes of FLASH memory — An erase operation erases all FLASH locations; a program operation programs all FLASH locations.
2. There must be no power outage during erase or program operations; otherwise, a high voltage must be applied to the $\overline{\text{IRQ1}}$ pin so that the MCU can enter Monitor mode. The alternative is to extract the MCU off the target system and reprogrammed using an external programmer.

Further cost-savings can be achieved by using the circuit in figure 4 to replace the SPGMR08.

Other ICP methods that can be applied to the JL3 are described in the application note: *AN1770 — In-Circuit Programming of FLASH memory in the MC68HC908GP20*.

Program Listing

```
;------
; Assembler Directives
; $base          10t
;------
; 68HC908JL3 User Mode FLASH Mass Erase
;
; Author       : Roger Fan
;
; File Name    : jl3icp.asm
;
;------
; Description:
; This program allows the MCU to mass erase itself in user mode.
; The detect condition for mass erase is PTB0=1 & PTD3=0.
;
; For successful code execution, the user should set a bus frequency of
; 2.4576MHz. This can be derived from a 9.8304MHz xtal for the HC908 part.
;
; The program uses a subroutine, erase_cmd, located at $FC06 in the
; monitor ROM, for the mass erase operation.
;
; Jumper setting during power-up reset:
;
;      Jumper          user mode          mass erase mode
;      ----          -          -
;      PTB0           -          pull-up(10k)
;      IRQ            pull-up          pull-up
;      PTD3           pull-up (10k)    short to ground
;------
; Version      Date          Description
;
; 0.1          06/10/99      Origin version
;------
; MCU (JL3) I/O pin Assignment
;------
PTA      equ    0          ; Port A
PTB      equ    1          ; Port B
PTD      equ    3          ; Port D
DDRA     equ    4          ; Port A direction register
DDRB     equ    5          ; Port B direction register
DDRD     equ    7          ; Port D direction register
s_data   equ    0          ; Serial data used in monitor mode
Ps_data  equ    PTB       ; Port location of serial data
DDRs_data equ    DDRB     ; Port direction location of serial data
;------
; FLASH Control Register
;------
FLCR     equ    $fe08     ; FLASH Control Register
HVEN     equ    3
MASS     equ    2
```



```

ERASE      equ    1
PGM        equ    0

FLBPR      equ    $fe09      ; FLASH Block Protect Register

;-----
; External Subroutine Call Declaration
;-----

erase_cmd  equ    $fc06      ; this routine is resident in the monitor rom,
                           ; and will erase an area unprotected when called

;-----
; Constant declaration
;-----

RAM_BEGIN  equ    $80        ; FLASH memory start address

;-----
; RAM declaration          ; required by erase subroutine in monitor ROM
;-----

ctrlbyt    equ    $88
cpuspd     equ    $89

CONFIG1    equ    $1F
CONFIG2    equ    $1E

RAM        equ    $90

MAIN       equ    $FB00
RSTVECTOR  equ    $FFFE

;-----
; Main Program
;-----

        org     MAIN

START:   rsp
        sei
        clr     CONFIG2
        mov     #$31,CONFIG1 ; disable COP & LVI
        clr     DDRB          ; check user mode mass erase condition
        clr     DDRD          ; PTB0=5V & PTD3=GND in user mode condition
        brclr  0,PTB,USERCODE ; check PTB0=5V
        brset  3,PTD,USERCODE ; check PTD3=GND

        clr     clrx

NEXTRAM:
        lda     MASS_ERASE,x ; Load mass erase code from FLASH to RAM
        sta     RAM,x
        incx
        cbeqx  #{ENDRAM-MASS_ERASE},RUNRAM
        bra     NEXTRAM

```

```

RUNRAM:
    jmp    RAM            ; Execute the mass erase

USERCODE:
    bra    *              ; Start of the user application code

;-----
; Mass Erase
;-----
MASS_ERASE:
    lda    #$ff          ; unprotect all FLASH area
    sta    FLBPR

    mov    #%01000000,ctrlbyt      ; setup mass erase
    mov    #10,cpuspd

    ldhx   #$ffff

    jsr    erase_cmd      ; mass erase routine


Mem_check
    ldx    #$0A
    lda    $FFF6,x
    cmp    #$FF
    bne    M_erase
    decx
    bne    Mem_check

ICPMODE:
    bra    *              ; Waiting for power-off the device,
                        ; then enter the ICP mode using SPGMR &
                        ; MCUscribe

M_erase
    jmp    RAM

ENDRAM:
    org    RSTVECTOR
    fdb    START          ; RESET

```

Motorola reserves the right to make changes without further notice to any products herein. Motorola makes no warranty, representation or guarantee regarding the suitability of its products for any particular purpose, nor does Motorola assume any liability arising out of the application or use of any product or circuit, and specifically disclaims any and all liability, including without limitation consequential or incidental damages. "Typical" parameters which may be provided in Motorola data sheets and/or specifications can and do vary in different applications and actual performance may vary over time. All operating parameters, including "Typicals" must be validated for each customer application by customer's technical experts. Motorola does not convey any license under its patent rights nor the rights of others. Motorola products are not designed, intended, or authorized for use as components in systems intended for surgical implant into the body, or other applications intended to support or sustain life, or for any other application in which the failure of the Motorola product could create a situation where personal injury or death may occur. Should Buyer purchase or use Motorola products for any such unintended or unauthorized application, Buyer shall indemnify and hold Motorola and its officers, employees, subsidiaries, affiliates, and distributors harmless against all claims, costs, damages, and expenses, and reasonable attorney fees arising out of, directly or indirectly, any claim of personal injury or death associated with such unintended or unauthorized use, even if such claim alleges that Motorola was negligent regarding the design or manufacture of the part. Motorola and  are registered trademarks of Motorola, Inc. Motorola, Inc. is an Equal Opportunity/Affirmative Action Employer.

How to reach us:

USA/EUROPE/Locations Not Listed: Motorola Literature Distribution; P.O. Box 5405, Denver, Colorado 80217. 1-800-441-2447 or 1-303-675-2140

JAPAN: Nippon Motorola Ltd. SPD, Strategic Planning Office 4-32-1, Nishi-Gotanda, Shinagawa-ku, Tokyo 141, Japan. 03-5487-8488

Mfax™, Motorola Fax Back System: RMFAX0@email.sps.mot.com; <http://sps.motorola.com/mfax/>; TOUCHTONE 1-602-244-6609;

US and Canada ONLY 1-800-774-1848

HOME PAGE: <http://motorola.com/sps/>

Mfax is a trademark of Motorola, Inc.

© Motorola, Inc., 1999



MOTOROLA
